
Subject: Re: Having trouble with code for data to image.
Posted by [David Fanning](#) on Wed, 26 Nov 2008 23:08:21 GMT
[View Forum Message](#) <> [Reply to Message](#)

mbweller@gmail.com writes:

> My image is of the western hemisphere of Mars, so lat range is -90 to
> 90 and lon range =3D -180 to 0 degrees. I would like to present it as a
> Hemisphere display since that makes the most sense to do.

You can try something like this:

```
image = BytScl(Dist(400), TOP=254)
CTLoad, 4, /Brewer, NCOLORS=254
TVLCT, FSC_Color('ivory', /TRIPLE), 255
Window, XSIZE=400, YSIZE=400
Erase, Color=FSC_Color('ivory')
Map_Set, /Orthographic, 0, -90, $
    Position=[0.1, 0.1, 0.9, 0.9], /NOERASE
warp = Map_Image(image, xs, ys, COMPRESS=1, $
    LATMIN=-90, LATMAX=90, LONMIN=-180, LONMAX=0, $
    MISSING=255)
TVIMAGE, warp, Position=[0.1, 0.1, 0.9, 0.9]
XYOuts, 0.5, 0.055, /NORMAL, '-90', $
    COLOR=FSC_Color('charcoal'), ALIGN=0.5
XYOuts, 0.5, 0.915, /NORMAL, '90', $
    COLOR=FSC_Color('charcoal'), ALIGN=0.5
XYOuts, 0.09, 0.5, /NORMAL, '-180', $
    COLOR=FSC_Color('charcoal'), ALIGN=1.0
XYOuts, 0.915, 0.5, /NORMAL, '0', $
    COLOR=FSC_Color('charcoal'), ALIGN=0.5
END
```

Cheers,

David

--

David Fanning, Ph.D.
Fanning Software Consulting, Inc.
Coyote's Guide to IDL Programming: <http://www.dfanning.com/>
Sepore ma de ni thui. ("Perhaps thou speakest truth.")
